

# Maple: An Introduction

August, 2000

\*\*\*\*\*

These notes are based on notes originally written by Dominik Gruntz, Kathrin Meier, and Michael Monagan of the Institute for Scientific Computation, ETH Zürich. They have been updated and modified for PHYCS 3310 in August, 2000, by Richard Price.

\*\*\*\*\*

## 1 Introduction

This introductory article contains basic information for the new user of Maple who wants to spend only a couple of hours learning Maple. This document covers how to use Maple as a calculator, recovering from errors, Maples notation for arithmetic, mathematical constants and functions, how to get help, how to solve equations, do linear algebra, and to define and plot mathematical functions in two and three dimensions. For each section there are several examples. At the end there are some exercises which you can try to solve. Note, we have not attempted to show any of Maples programming facilities in this article.

## 2 Getting Started

The way of opening Maple will be somewhat different on a PMT PC running Windows and or running Linux, or a Physics Department workstation. But no matter how it is invoked you will find yourself looking at a Windows-like window with menus and a toolbar of icons along the top border. (Hint: If you go to the “Help” menu in the upper right corner, you can click on “Balloon Help,” which will enable comments that tell you the action of the icons in the toolbar.)

In the open space beneath the toolbar you will see a  $>$ , which is the prompt used in Maple, the way input is started. The next important thing to learn is how input is ended.

<i>ATTENTION</i> : every Maple command must end with a semicolon ;
--

This is a grammatical requirement of the Maple language. The semicolon tells Maple that this is the end of the input. Type now `4/6;` followed by a return. Maple answers with

$2/3$

To refer to the latest result, one uses the percentage character `.` Try it now

`> %;`

$2/3$

You can use the value of `%` in an expression as if it were a variable, for example

`> 2*%+1;`

$7/3$

In order to save results, one assigns them to a variable, e.g.

`> R := %;`

$R := 7/3$

### 3 Arithmetic and Maple Notation

The basic arithmetic operators and constants known to Maple are

Maple notation	Meaning	Mathematical notation
<code>x+y</code>	addition	$+$
<code>-x</code> and <code>x-y</code>	negation and subtraction	$-x$ and $x - y$
<code>x*y</code>	multiplication	$xy$
<code>x/y</code>	division	$\frac{x}{y}$
<code>x^y</code> or <code>x**y</code>	exponentiation	$x^y$
<code>k!</code>	factorial	$k!$
<code>I</code> or <code>sqrt(-1)</code>	complex unit	$i$ or $\sqrt{-1}$
<code>Pi</code>		$\pi$
<code>infinity</code>		$\infty$
<code>abs(x)</code>	absolute value	$ x $
<code>sqrt(x)</code> or <code>x^(1/2)</code>	square root	$\sqrt{x}$
<code>exp(x)</code>	exponential function	$e^x$
<code>ln(x)</code> or <code>log(x)</code>	natural logarithm	$\ln x$
<code>sin(x)</code>	sine function	$\sin x$
<code>cos(x)</code>	cosine function	$\cos x$
<code>tan(x)</code>	tangent function	$\tan x$

Notice that Maple does arithmetic with integers exactly. That is, exact arithmetic is used instead of decimal arithmetic. Use decimal numbers if you want decimal numbers. Examples

```
> 2*3+2/7;
                                         44/7
```

```
> 2.0*3.0+2.0/7;
                                         6.285714286
```

This principle works for formulae too. Use the `evalf` function if you want a decimal approximation.

```
> sin(Pi/3);
                                         1/2
                                         1/2 3
```

```
> evalf(%);
                                         .8660254040
```

There is no limit on the length of integers in Maple. It is quite common to compute with integers several hundred digits long. It is also possible to do decimal arithmetic to more than the default 10 digits of precision. You can compute  $\pi$  to a 1000 digits if you want. This is done by assigning the the global variable `Digits` to the desired precision. Note, don't forget to reset `Digits` to 10 if you don't need more than 10 Digits! High precision operations take longer! Here are some examples

```
> 2^100;
                                         1267650600228229401496703205376
```

```
> Digits := 50;
> evalf( sin(Pi/3) );
                                         .86602540378443864676372317075293618347140262690520
```

```
> Digits := 10;
```

In the above examples we have used the colon to terminate a command. Use the colon `:` instead of a semicolon `;` if you don't want to see the output.

## 4 Mistakes and Errors

If you forget the semicolon don't panic! Maple will simply print another prompt indicating that it is still waiting for more input. Just type the semicolon and return and Maple will go ahead and compute the result. Note, this means that you can enter large expressions over one or more lines. Try to input a large expression like the following

```
> f := 4*x^4+3*x^3*y+2*x^2*y^2+
>      3*x*y^3+4*y^4;
```

$$f := 4 x^4 + 3 x^3 y + 2 x^2 y^2 + 3 x y^3 + 4 y^4$$

If you incorrectly input an expression, Maple will respond with a *syntax error*. A common error is to forget the semicolon and then to try to correct the problem by retyping the input. For example, if you type

```
> 2*x+1
> 2*x+1
```

What you typed is equivalent to typing

```
> 2*x+1 2*x+1
```

Maple will give you an error message. (In Maple version V, the cursor will go to the end of the first line, and the message “unexpected number” will appear, indicating that Maple does not understand what you mean by 1 followed by a space and a 2.) To recover from a syntax error, just type a semicolon to clear the remaining input, and then reenter the command.

## 5 On-Line Help

If you need help for a Maple function, you can use the `?` command, which opens a separate help window. If, for example, you wanted help on the Maple command for cosine, you would type

```
> ?cos
```

Note that there is no semicolon or colon at the end of the line. After typing the help request just do a return. Try this now, e.g., for the maple function `min`.

## 6 Simplification Commands

The output from Maple is not always in the simplest form. There are a number of commands such as `collect`, `combine`, `expand`, `factor` and `simplify` which can be used to simplify or rearrange the output into the desired form. For example, given the polynomial

```
> (x+y)*(x-y)-x^2;
```

$$(x + y)(x - y) - x^2$$

if we multiply it out with the `expand` command the result is simpler

```
> expand(%);
```

$$-y^2$$

Often factoring a polynomial yields a simpler result, e.g.

```
> x^4+x^2*y+2*x^2+2*x^3+2*x*y+2*x+y+1;
```

$$x^4 + x^2 y + 2 x^2 + 2 x^3 + 2 x y + 2 x + y + 1$$

```
> factor(%);
```

$$(x + 1)^2 (x^2 + y + 1)$$

The `simplify` command can simplify general expressions, e.g.

```
> simplify(exp(a+ln(b*exp(c))));
```

$$be^{(a+c)}$$

In this example the `simplify` command reduces a rational function

```
> (x^3-y^3)/(x^2+x*y+y^2);
```

$$\frac{x^3 - y^3}{x^2 + x y + y^2}$$

```
> simplify(%);
```

$$x - y$$

## 7 Calculus Commands

Experiment with the integration, differentiation and summation commands. The Maple syntax for these functions and examples is as follows.

Maple notation	Meaning	Mathematical notation
<code>diff(f(x),x);</code>	Derivative	$\frac{d}{dx}f(x)$
<code>int(f(x),x);</code>	Indefinite integration	$\int f(x)dx$
<code>sum(f(x),x);</code>	Indefinite summation	$\sum_n f(n)$
<code>int(f(x),x=a..b);</code>	Definite integration	$\int_a^b f(x)dx$
<code>sum(f(k),k=a..b);</code>	Definite summation	$\sum_{k=a}^b f(k)$

```
> diff(ln(x),x);
```

$$1/x$$

```
> diff(arcsin(a*x),x);
```

$$\frac{a}{(1 - a^2 x^2)^{1/2}}$$

```
> int(x*ln(x),x);
```

$$\frac{1}{2} x^2 \ln(x) - \frac{1}{4} x^2$$

```
> int(sqrt(1-x^2),x);
```

$$\frac{1}{2} x \sqrt{1 - x^2} + \frac{1}{2} \arcsin(x)$$

```
> int(ln(x),x=1..2);
```

```

- 1 + 2 ln(2)
> sum(n^2,n);
      3      2
1/3 n  - 1/2 n  + 1/6 n
> sum(a^k,k=1..n);
      (n + 1)
      a
-----
      a - 1
      a
-----
      a - 1

```

In the case of indefinite integration, you can test whether Maple's answer is right by differentiating the integral and subtracting it from the integrand. The difference should be zero! But Maple may not recognize that the difference is zero immediately. You may have to help Maple simplify the difference to 0. You can make use of `expand` and `simplify`. Example:

```

> f := (x^3+2*x^2-x)/(x^3-x^2+x-1);
      3      2
      x  + 2 x  - x
f := -----
      3      2
      x  - x  + x - 1
> int(f,x);
      2
x + ln(x - 1) + ln(x  + 1)
> diff(%,x);
      1      x
1 + ----- + 2 -----
      x - 1      2
              x  + 1
> simplify(%-f);
0

```

## 8 The Solve Commands

You can solve single equations and systems of linear and non-linear equations exactly with the `solve` command. For approximate solutions, use the `fsolve` command. For ordinary differential equations use the `dsolve` command. And for recurrence equations use the `rsolve` command. These commands have the following syntax. For more detail, use `?solve` and so forth.

<code>solve(e, x)</code>	solve the equation $e$ for the unknown $x$
<code>solve({e1, ..., en}, {x1, ..., xm})</code>	solve the equations $e_1, \dots, e_n$ for $x_1, \dots, x_m$
<code>fsolve(e, x)</code>	solve the equation $e$ numerically for $x$
<code>fsolve({e1, ..., en}, {x1, ..., xm})</code>	solve numerically $e_1, \dots, e_n$ for $x_1, \dots, x_m$
<code>dsolve(e, y(x))</code>	solve the ODE $e$ for $y(x)$
<code>dsolve({e, i1, ..., in}, y(x))</code>	solve the ODE $e$ and initial conditions $i_1, \dots, i_n$ for $y(x)$
<code>rsolve(e, f(n))</code>	solve the recurrence equation $e$ for $f(n)$
<code>rsolve({e, i1, ..., in}, f(n))</code>	solve the recurrence $e$ given initial conditions $i_1, \dots, i_n$ for $f(n)$

For example, to solve the equation  $x^3 - 6x = 5$  for  $x$

```
> solve( x^3-6*x=5, x );
```

$$-1, \frac{1}{2} + \frac{1}{2} \sqrt{21}, \frac{1}{2} - \frac{1}{2} \sqrt{21}$$

To solve the ODE (ordinary differential equation)  $y(x) + 2y''(x) = e^x$  given initial conditions  $y(0) = 1$  and  $y'(0) = 0$

```
> dsolve({y(x)+2*diff(y(x),x$2)=exp(x), y(0)=1, D(y)(0)=0}, y(x));
```

This will probably (depending on the Maple version and platform) give a rather complicated-looking output, but it can be simplified to give

```
> simplify(%);
```

$$y(x) = \frac{1}{3} \exp(x) + \frac{2}{3} \cos\left(\frac{1}{2} \sqrt{2} x\right) - \frac{1}{3} \sin\left(\frac{1}{2} \sqrt{2} x\right) \sqrt{2}$$

## 9 Other Commands and Functions

Maple knows about the elementary functions `ln`, `exp`, `sin`, `cos`, `tan`, etc. Some other special functions that you might need are

Maple notation	Function	Mathematical notation
<code>erf(x)</code>	Error function	$erf(x)$
<code>binomial(n,k)</code>	Binomial function	$\binom{n}{k}$
<code>GAMMA(x)</code>	Gamma function	$\Gamma(x)$
<code>LegendreP(n,x)</code>	Legendre function	$P_n(x)$

Note that the command `LegendreP(3,2.7)` will generate the numerical value of  $P_3(2.7)$ , but it cannot be used to generate the form of the polynomial  $P_3(x)$ . To do this you need to open a Maple package. (For a list of Maple packages type `> ?packages`. The package you need for Legendre polynomials is the `orthopoly` package. Try the following

```
> with(orthopoly);  
> P(n,x);
```

The output will be the  $n = 3$  Legendre polynomial  $(5x^3 - 3x)/2$ .

## 10 Data Structures

Maple uses sequences, lists, sets, tables and arrays for representing more complicated data. See `?sequences`, `?list`, `?set`, `?table`, `?array` for more detailed help and examples. A sequence is a sequence of expressions separated by commas. The `seq` function is a very useful function for creating sequences. Its syntax is

```
seq( f(i), i=a..b )
```

Lists group together the elements of a sequence; they are created using square brackets `[ , ]`. Sets are like lists except duplicate entries are removed. Sets are created using squiggly brackets `{ , }`.

Some examples:

```
> S1 := 1,5,3;
                                     S1 := 1, 5, 3

> max(S1);
                                     5

> S2 := seq(i^2, i=1..5);
                                     S2 := 1, 4, 9, 16, 25

> S2[3];
                                     9

> S1 := {x,y,z,y};
                                     S1 := {x, y, z}

> S1 union {w,x};
                                     {x, y, z, w}

> L1 := [x,y,z,y];
                                     L1 := [x, y, z, y]

> L2 := [L1[4],L1[2..3]];
                                     L2 := [y, y, z]
```

## 11 The Linear Algebra Package

In Maple a vector is represented as a one-dimensional array, and a matrix is represented as a two-dimensional array. See `?vector` and `?matrix` for detailed help on vectors and matrices.

In Maple there are many packages for special applications. The `linalg` package contains many functions from linear algebra for computing with vectors and matrices. In order to use a package, you must load the package using the `with` command, e.g.

```
> with(linalg);
Warning: new definition for   norm
Warning: new definition for   trace
[BlockDiagonal, GramSchmidt, JordanBlock, add, addcol, addrow, adj, adjoint,
 angle, augment, backsub, band, basis, bezout, charmat, charpoly, col,
 coldim, colspace, colspan, companion, concat, cond, copyinto, crossprod,
 curl, definite, delcols, delrows, det, diag, diverge, dotprod, eigenvals,
 eigenvects, equal, exponential, extend, ffgausselim, fibonacci, frobenius,
 gausselim, gaussjord, genmatrix, grad, hadamard, hermite, hessian,
 hilbert, htranspose, ihermite, indexfunc, innerprod, intbasis, inverse,
 ismith, iszero, jacobian, jordan, kernel, laplacian, leastsqrs, linsolve,
 matrix, minor, minpoly, mulcol, mulrow, multiply, norm, nullspace, orthog,
 permanent, pivot, potential, randmatrix, range, rank, row, rowdim,
 rowspan, rref, scalarmul, singularvals, smith, stack, submatrix,
 subvector, subbasis, swapcol, swaprow, sylvester, toeplitz, trace,
 transpose, vandermonde, vecpotent, vectdim, vector]
```

You can now use any of the functions listed. The `matrix` command can be used to input a matrix. In the next example the inverse and determinant of a 3 by 3 matrix is computed.

```
> A := matrix([[x-1,2,3],[0,x-2,2],[2,1,x-3]]);
```

$$A := \begin{pmatrix} x-1 & 2 & 3 \\ 0 & x-2 & 2 \\ 2 & 1 & x-3 \end{pmatrix}$$

```
> det(A);
```

$$x^3 - 6x^2 + 3x + 16$$

```
> inverse(A);
```

$$\begin{pmatrix} \frac{x^2-5x+4}{x^3-6x^2+3x+16} & -\frac{2x-9}{x^3-6x^2+3x+16} & -\frac{3x-10}{x^3-6x^2+3x+16} \\ \frac{4}{x^3-6x^2+3x+16} & \frac{x^2-4x-3}{x^3-6x^2+3x+16} & -\frac{2x-2}{x^3-6x^2+3x+16} \\ -\frac{2x-4}{x^3-6x^2+3x+16} & -\frac{x-5}{x^3-6x^2+3x+16} & \frac{(x-1)(x-2)}{x^3-6x^2+3x+16} \end{pmatrix}$$

## 12 Functions and Graphics

Mathematical functions of one or more variables can be defined in Maple. The function can be evaluated at either numerical or symbolic expressions. For example

```
> f:=x->sin(x)/x;
```

$$f := x \rightarrow \frac{\sin(x)}{x}$$

```
> f(2.0);
```

$$.4546487134$$

```
> f(t);
```

$$\frac{\sin(t)}{t}$$

Functions can be graphed with the `plot` command which has the following syntax.

`plot( f, a..b )`

That means, that the function  $f$  will be drawn on the interval  $a$  to  $b$ . Try this now  

```
> plot(f,-12..12);
```

An example of a function in two variables

```
> g:=(x,y)->(x^2-y^2)/(x^2+y^2);
```

$$g := (x,y) \rightarrow \frac{x^2 - y^2}{x^2 + y^2}$$

```
> g(1,2);
```

-3/5

```
> g(1,x);
```

$$\frac{1 - x^2}{1 + x^2}$$

Functions of two variables can be graphed with the `plot3d` command

```
plot3d( f, a..b, c..d )
```

```
> plot3d(g, -1..1, -1..1);
```

The examples above are plots of functions. The other possibility is to graph an *expression*. The syntax and examples for plotting expressions is

```
plot( f(x), x=a..b )  
plot3d( f(x,y), x=a..b, y=c..d )
```

```
> plot( sin(x)/x, x=-12..12 );
```

```
> plot3d( (x^2-y^2)/(x^2+y^2), x=-1..1, y=-1..1 );
```

## Exercises

(PHYCS 3910 students, note: The following exercises were included by the authors of these notes; they are not assigned problems for the course.)

1. Calculate the first and second derivative of  $\sin(x) \cos(x)$  with respect to  $x$ .
2. Given the polynomial  $y(x) = x^3 - 4x^2 + 4x - 1$  find the roots and any local minima and maxima of  $y(x)$ . Check your solutions with a plot of the polynomial.
3. Given  $f = x^2 - 4$  calculate the integral of  $f$  and  $\frac{1}{f}$  with respect to  $x$ . Check that Maples answers are correct by differentiating the results.
4. Compute the following integrals

$$\int_0^{\infty} e^{-t} dt \quad \text{and} \quad \int_0^{\infty} e^{-t^2} dt$$

5. Compute the following sums

$$\sum_{k=1}^{1000} k \quad \text{and} \quad \sum_{k=1}^{\infty} 1/k^2$$

6. Compute a formula for the sum of the first  $n$  integers and also the sum of the squares of the first  $n$  integers. I.e. compute and simplify the sums

$$\sum_{k=1}^n k \quad \text{and} \quad \sum_{k=1}^n k^2$$

7. Given the function  $h(x) = 1 - x + \sin(x)$ , define the function in Maple and compute the value of the function  $h$  at  $x = -2.0$  and then graph the function in the region  $[-5,5]$ .
8. Input the following Matrix  $A$  into Maple

$$A = \begin{pmatrix} a & 0 & 5 \\ 1 & 1 & 1 \\ -a & 0 & 0 \end{pmatrix}$$

- (a) Compute the characteristic polynomial of  $A$ . Hint: use the `charmat` and `det` commands in the linear algebra package `linalg`.
  - (b) Compute the eigenvalues of  $A$ . Hint: use the `solve` or `factor` commands to find the roots of the characteristic polynomial.
9. Use the `solve` command to solve the linear system

$$\begin{cases} 4x - 5y = 11 \\ 2x + y = 9 \end{cases}$$